Pensieve header: The full sl_2 invariant using the Drinfel'd double. Based on Projects/SL2Invariant/SL2Invariant.nb.

Program

Program

Utilities

In[∘]:=
```
$k = 2; (*ℏ=γ=1;*)
```

Canonical Form:

Program

In[∘]:=
```
CF[sd_SeriesData] := MapAt[CF, sd, 3];
CF[ℰ_] := ExpandDenominator@ExpandNumerator@Together[
    Expand[ℰ] //. e^x_ e^y_ :> e^(x+y) /. e^x_ :> e^CF[x]];
```

The Kronecker δ:

Program

In[∘]:=
```
Kδ /: Kδ_{i_,j_} := If[i === j, 1, 0];
```

Equality, multiplication, and degree-adjustment of perturbed Gaussians; $\mathbb{E}[L, Q, P]$ stands for $e^{L+Q} P$:

Program

In[∘]:=
```
𝔼 /: 𝔼[L1_, Q1_, P1_] ≡ 𝔼[L2_, Q2_, P2_] :=
  CF[L1 == L2] ∧ CF[Q1 == Q2] ∧ CF[Normal[P1 - P2] == 0];
𝔼 /: 𝔼[L1_, Q1_, P1_] 𝔼[L2_, Q2_, P2_] := 𝔼[L1 + L2, Q1 + Q2, P1 * P2];
𝔼[L_, Q_, P_]_{$k_} := 𝔼[L, Q, Series[Normal@P, {ϵ, 0, $k}]];
```

Program

Zip and Bind

Variables and their duals:

Program

In[∘]:=
```
{t*, b*, y*, a*, x*, z*} = {τ, β, η, α, ξ, ζ};
{τ*, β*, η*, α*, ξ*, ζ*} = {t, b, y, a, x, z}; (u_{i_})* := (u*)_i;
```

Finite Zips:

Program

In[∘]:=
```
collect[sd_SeriesData, ζ_] := MapAt[collect[#, ζ] &, sd, 3];
collect[ℰ_, ζ_] := Collect[ℰ, ζ];
Zip_{}[P_] := P; Zip_{ζ_,ζs___}[P_] :=
 (collect[P // Zip_{ζs}, ζ] /. f_. ζ^d_. :> ∂_{ζ*,d} f) /. ζ* → 0
```

Zip

```
Zip_{ζs_List}@𝔼[Q_, P_] := Module[{ζ, z, zs, c, ys, ηs, qt, zrule, ζrule},
   zs = Table[ζ*, {ζ, ζs}];
   c = Q /. Alternatives @@ (ζs ⋃ zs) → 0;
   ys = Table[∂_ζ (Q /. Alternatives @@ zs → 0), {ζ, ζs}];
   ηs = Table[∂_z (Q /. Alternatives @@ ζs → 0), {z, zs}];
   qt = Inverse@Table[Kδ_{z,ζ*} − ∂_{z,ζ}Q, {ζ, ζs}, {z, zs}];
   zrule = Thread[zs → qt.(zs + ys)];
   ζrule = Thread[ζs → ζs + ηs.qt];
   Simplify /@ 𝔼[c + ηs.qt.ys, Det[qt] Zip_{ζs}[P /. (zrule ⋃ ζrule)]] ];
```

In[∘]:=

```
HL[ℰ_] := Style[ℰ, Background → Yellow];
```

In[∘]:=

```
Eh = 𝔼[h Sum[a_{10 i+j} x_i ξ_j, {i,1,3}, {j,1,3}], Sum[f_i[x_1, x_2, x_3] ξ_i, {i,1,3}]]; E1 = Eh /. h → 1;
Short[lhs = Zip_{ξ1,ξ2}@E1, 5]
HL[lhs == Zip_{ξ1}@Zip_{ξ2}@E1 == Zip_{ξ2}@Zip_{ξ1}@E1]
```

Out[∘]//Short= $\mathbb{E}\left[a_{11} + a_{22} + a_{33}\, x_3\, \xi_3,\ \xi_3\, f_3[0, 0, x_3] + f_2^{(0,1,0)}[0, 0, x_3] + f_1^{(1,0,0)}[0, 0, x_3]\right]$

Out[∘]= True

QZip implements the "Q-level zips" on $\mathbb{E}(L, Q, P) = Pe^{L+Q}$. Such zips regard the L variables as scalars.

Program

In[∘]:=

```
QZip_{ζs_List}@𝔼[L_, Q_, P_] := Module[{ζ, z, zs, c, ys, ηs, qt, zrule, ζrule},
   zs = Table[ζ*, {ζ, ζs}];
   c = CF[Q /. Alternatives @@ (ζs ⋃ zs) → 0];
   ys = CF@Table[∂_ζ (Q /. Alternatives @@ zs → 0), {ζ, ζs}];
   ηs = CF@Table[∂_z (Q /. Alternatives @@ ζs → 0), {z, zs}];
   qt = CF@Inverse@Table[Kδ_{z,ζ*} − ∂_{z,ζ}Q, {ζ, ζs}, {z, zs}];
   zrule = Thread[zs → CF[qt.(zs + ys)]];
   ζrule = Thread[ζs → ζs + ηs.qt];
   CF /@ 𝔼[L, c + ηs.qt.ys, Det[qt] Zip_{ζs}[P /. (zrule ⋃ ζrule)]] ];
```

Upper to lower and lower to Upper:

Program

In[∘]:=

```
U2l = {B_{i_}^{p_.} → e^{-p ħ γ b_i}, B^{p_.} → e^{-p ħ γ b}, T_{i_}^{p_.} → e^{p ħ t_i}, T^{p_.} → e^{p ħ t}, 𝒜_{i_}^{p_.} → e^{p γ α_i}, 𝒜^{p_.} → e^{p γ α}};
l2U = {e^{c_. b_{i_} + d_.} :> B_i^{-c/(ħ γ)} e^d, e^{c_. b + d_.} :> B^{-c/(ħ γ)} e^d,
   e^{c_. t_{i_} + d_.} :> T_i^{c/ħ} e^d, e^{c_. t + d_.} :> T^{c/ħ} e^d,
   e^{c_. α_{i_} + d_.} :> 𝒜_i^{c/γ} e^d, e^{c_. α + d_.} :> 𝒜^{c/γ} e^d,
   e^{ℰ_} :> e^{Expand@ℰ}};
```

LZip implements the "L-level zips" on $\mathbb{E}(L, Q, P) = Pe^{L+Q}$. Such zips regard all of Pe^Q as a single"P". Here the z's are b and α and the ζ's are β and a.

Program

```
In[∘]:=
LZip_{ζs_List}@𝔼[L_, Q_, P_] := Module[{ζ, z, zs, c, ys, ηs, lt, zrule, L1, L2, Q1, Q2},
   zs = Table[ζ*, {ζ, ζs}];
   c = L /. Alternatives @@ (ζs ⋃ zs) → 0;
   ys = Table[∂_ζ (L /. Alternatives @@ zs → 0), {ζ, ζs}];
   ηs = Table[∂_z (L /. Alternatives @@ ζs → 0), {z, zs}];
   lt = Inverse@Table[Kδ_{z,ζ*} - ∂_{z,ζ} L, {ζ, ζs}, {z, zs}];
   zrule = Thread[zs → lt.(zs + ys)];
   L2 = (L1 = c + ηs.zs /. zrule) /. Alternatives @@ zs → 0;
   Q2 = (Q1 = Q /. U2l /. zrule) /. Alternatives @@ zs → 0;
   CF /@ 𝔼[L2, Q2, Det[lt] e^{-L2-Q2} Zip_{ζs}[e^{L1+Q1} (P /. U2l /. zrule)]] //. l2U];
```

Program

```
In[∘]:=
B_{}[L_, R_] := L R;
B_{{is__}}[L_𝔼, R_𝔼] := Module[{n}, Times[
      L /. Table[(v : b | B | t | T | a | x | y)_i → v_{n@i}, {i, {is}}],
      R /. Table[(v : β | τ | α | 𝒜 | ξ | η)_i → v_{n@i}, {i, {is}}]
     ] // LZip_{Join@@Table[{β_{n@i},τ_{n@i},a_{n@i}},{i,{is}}]} // QZip_{Join@@Table[{ξ_{n@i},y_{n@i}},{i,{is}}]} ];
B_{is___}[L_, R_] := B_{{is}}[L, R];
```

Program

𝔼 morphisms with domain and range.

Program

```
In[∘]:=
B_{is_List}[𝔼_{d1_→r1_}[L1_, Q1_, P1_], 𝔼_{d2_→r2_}[L2_, Q2_, P2_]] :=
  𝔼_{(d1⋃Complement[d2,is])→(r2⋃Complement[r1,is])} @@ B_{is}[𝔼[L1, Q1, P1], 𝔼[L2, Q2, P2]];
𝔼_{d1_→r1_}[L1_, Q1_, P1_] // 𝔼_{d2_→r2_}[L2_, Q2_, P2_] :=
  B_{r1⋂d2}[𝔼_{d1→r1}[L1, Q1, P1], 𝔼_{d2→r2}[L2, Q2, P2]];
𝔼_{d1_→r1_}[L1_, Q1_, P1_] ≡ 𝔼_{d2_→r2_}[L2_, Q2_, P2_] ^:=
  (d1 == d2) ∧ (r1 == r2) ∧ (𝔼[L1, Q1, P1] ≡ 𝔼[L2, Q2, P2]);
𝔼_{d1_→r1_}[L1_, Q1_, P1_] 𝔼_{d2_→r2_}[L2_, Q2_, P2_] ^:=
  𝔼_{(d1⋃d2)→(r1⋃r2)} @@ (𝔼[L1, Q1, P1] 𝔼[L2, Q2, P2]);
𝔼_{d_→r_}[L_, Q_, P_]_{$k_} := 𝔼_{d→r} @@ 𝔼[L, Q, P]_{$k};
𝔼_[ℰ___][i_] := {ℰ}⟦i⟧;
```

Program

“Define” code

Define[lhs = rhs, ...] defines the lhs to be rhs, except that rhs is computed only once for each value of $k. Fancy Mathematica not for the faint of heart. Most readers should ignore.

Program

```
In[∘]:=
SetAttributes[Define, HoldAll];
Define[def_, defs__] := (Define[def]; Define[defs];);
```

Program

In[ ]:=
```
Define[op_is__ = ℰ_] := Module[{SD, ii, jj, kk, isp, nis, nisp, sis}, Block[{i, j, k},
   ReleaseHold[Hold[
      SD[op_{nisp,$k_Integer}, Block[{i, j, k}, op_{isp,$k} = ℰ; op_{nis,$k}]];
      SD[op_isp, op_{{is},$k}]; SD[op_{sis__}, op_{{sis}}];
     ] /. {SD → SetDelayed,
      isp → {is} /. {i → i_, j → j_, k → k_},
      nis → {is} /. {i → ii, j → jj, k → kk},
      nisp → {is} /. {i → ii_, j → jj_, k → kk_}
     }] ]]
```

Program

The Fundamental Tensors

Program

In[ ]:=
```
Define[am_{i,j→k} = 𝔼_{{i,j}→{k}}[(α_i + α_j) a_k, (e^{-γ α_j} ξ_i + ξ_j) x_k, 1]_{$k},
 bm_{i,j→k} = 𝔼_{{i,j}→{k}}[(β_i + β_j) b_k, (η_i + η_j) y_k, e^{(e^{-ϵ β_i} - 1) η_j y_k}]_{$k}]
```

Program

In[ ]:=
```
Define[R_{i,j} =
  𝔼_{{}→{i,j}}[ħ a_j b_i, ħ x_j y_i, e^(Sum_{k=2}^{$k+1} ((1 - e^{γ ϵ ħ})^k (ħ y_i x_j)^k) / (k (1 - e^{k γ ϵ ħ})))]_{$k}]
```

Program

In[ ]:=
```
Define[R̄_{i,j} = 𝔼_{{}→{i,j}}[-ħ a_j b_i, -ħ x_j y_i / B_i,
   1 + If[$k == 0, 0, (R̄_{{i,j},$k-1})_{$k}[3] -
      (((R̄_{{i,j},0})_{$k} R_{1,2} (R̄_{{3,4},$k-1})_{$k}) // (bm_{i,1→i} am_{j,2→j}) // (bm_{i,3→i} am_{j,4→j})) [3]]],
 P_{i,j} = 𝔼_{{i,j}→{}}[β_i α_j / ħ, η_i ξ_j / ħ, 1 + If[$k == 0, 0, (P_{{i,j},$k-1})_{$k}[3] -
      (R_{1,2} // ((P_{{1,j},0})_{$k} (P_{{i,2},$k-1})_{$k}))[3]]]]
```

Program

In[ ]:=
```
Define[aS_j = R̄_{i,j} ~ B_i ~ P_{i,j},
 aS̄_i = 𝔼_{{i}→{i}}[-a_i α_i, -x_i 𝒜_i ξ_i, 1 + If[$k == 0, 0, (aS̄_{{i},$k-1})_{$k}[3] -
      ((aS̄_{{i},0})_{$k} ~ B_i ~ aS_i ~ B_i ~ (aS̄_{{i},$k-1})_{$k}) [3]]]]
```

Program

In[ ]:=
```
Define[bS_i = R_{i,1} ~ B_1 ~ aS_1 ~ B_1 ~ P_{i,1},
 bS̄_i = R_{i,1} ~ B_1 ~ aS̄_1 ~ B_1 ~ P_{i,1},
 aΔ_{i→j,k} = (R_{1,j} R_{2,k}) // bm_{1,2→3} // P_{3,i},
 bΔ_{i→j,k} = (R_{j,1} R_{k,2}) // am_{1,2→3} // P_{i,3}]
```

Program

In[∘]:=

```
Define[dm_{i,j→k} = (𝔼_{{i,j}→{i,j}}[β_i b_i + α_j a_j, η_i y_i + ξ_j x_j, 1]
    (aΔ_{i→1,2} // aΔ_{2→2,3} // \overline{aS}_3) (bΔ_{j→-1,-2} // bΔ_{-2→-2,-3})) // (P_{-1,3} P_{-3,1} am_{2,j→k} bm_{i,-2→k}),
  dS_i = 𝔼_{{i}→{1,2}}[β_i b_1 + α_i a_2, η_i y_1 + ξ_i x_2, 1] // (\overline{bS}_1 aS_2) // dm_{2,1→i},
  dΔ_{i→j,k} = (bΔ_{i→3,1} aΔ_{i→2,4}) // (dm_{3,4→k} dm_{1,2→j})]
```

Program

In[∘]:=

```
Define[C_i = 𝔼_{{}→{i}}[0, 0, B_i^{1/2} e^{-ħ ϵ a_i/2}]_{$k},
  \overline{C}_i = 𝔼_{{}→{i}}[0, 0, B_i^{-1/2} e^{ħ ϵ a_i/2}]_{$k},
  Kink_i = (R_{1,3} \overline{C}_2) // dm_{1,2→1} // dm_{1,3→i},
  \overline{Kink}_i = (\overline{R}_{1,3} C_2) // dm_{1,2→1} // dm_{1,3→i}]
```

Note. $t == \epsilon a - \gamma b$ and $b == -t/\gamma + \epsilon a/\gamma$.

Program

In[∘]:=

```
Define[b2t_i = 𝔼_{{i}→{i}}[α_i a_i - β_i t_i / γ, ξ_i x_i + η_i y_i, e^{ϵ β_i a_i/γ}]_{$k},
  t2b_i = 𝔼_{{i}→{i}}[α_i a_i - τ_i γ b_i, ξ_i x_i + η_i y_i, e^{ϵ τ_i a_i}]_{$k}]
```

Program

In[∘]:=

```
Define[kR_{i,j} = R_{i,j} // (b2t_i b2t_j) /. t_{i|j} → t,
  \overline{kR}_{i,j} = \overline{R}_{i,j} // (b2t_i b2t_j) /. {t_{i|j} → t, T_{i|j} → T},
  km_{i,j→k} = (t2b_i t2b_j) // dm_{i,j→k} // b2t_k /. {t_k → t, T_k → T, τ_{i|j} → 0},
  kC_i = C_i // b2t_i /. T_i → T, \overline{kC}_i = \overline{C}_i // b2t_i /. T_i → T,
  kKink_i = Kink_i // b2t_i /. {t_i → t, T_i → T},
  \overline{kKink}_i = \overline{Kink}_i // b2t_i /. {t_i → t, T_i → T}]
```

Testing

In[∘]:=
```
Block[{$k = 1}, {
   am → am_{i,j→k}, bm → bm_{i,j→k}, dm → dm_{i,j→k}, R → R_{i,j}, R̄ → R̄_{i,j}, P → P_{i,j},
   aS → aS_i, aS̄ → aS̄_i, bS → bS_i, bS̄ → bS̄_i, dS → dS_i, aΔ → aΔ_{i→j,k}, bΔ → bΔ_{i→j,k},
   dΔ → dΔ_{i→j,k}, C → C_i, C̄ → C̄_i, Kink → Kink_i, Kink̄ → Kink̄_i, b2t → b2t_i, t2b → t2b_i
  }] //
 Column
```

Out[∘]=

$$\text{am} \to \mathbb{E}_{\{i,j\}\to\{k\}}\left[a_k\,(\alpha_i+\alpha_j),\ x_k\left(e^{-\gamma\,\alpha_j}\,\xi_i+\xi_j\right),\ 1\right]$$

$$\text{bm} \to \mathbb{E}_{\{i,j\}\to\{k\}}\left[b_k\,(\beta_i+\beta_j),\ y_k\,(\eta_i+\eta_j),\ 1-y_k\,\beta_i\,\eta_j\,\epsilon+O[\epsilon]^2\right]$$

$$\text{dm} \to \mathbb{E}_{\{i,j\}\to\{k\}}\Big[a_k\,\alpha_i+a_k\,\alpha_j+b_k\,\beta_i+b_k\,\beta_j,\ \frac{1}{\hbar\,\mathcal{A}_i\,\mathcal{A}_j}\left(\hbar\,y_k\,\mathcal{A}_i\,\mathcal{A}_j\,\eta_i+\hbar\,y_k\,\mathcal{A}_j\,\eta_j+\hbar\,x_k\,\mathcal{A}_i\,\xi_i+\mathcal{A}_i\,\mathcal{A}_j\,\eta_j\,\xi_i-B_k\,\mathcal{A}_i\,\mathcal{A}_j\,\eta_j\,\xi_i+\hbar\,x_k\,\mathcal{A}_i\,\mathcal{A}_j\,\xi_j\right),$$
$$1+\frac{1}{4\,\hbar\,\mathcal{A}_i\,\mathcal{A}_j}\big(-4\,\hbar\,y_k\,\mathcal{A}_j\,\beta_i\,\eta_j-4\,\hbar\,x_k\,\mathcal{A}_i\,\beta_j\,\xi_i+4\,\gamma\,\hbar^2\,x_k\,y_k\,\eta_j\,\xi_i+4\,\hbar\,a_k\,B_k\,\mathcal{A}_i\,\mathcal{A}_j\,\eta_j\,\xi_i+2\,\gamma\,\hbar\,y_k\,\mathcal{A}_j\,\eta_j^2\,\xi_i-6\,\gamma\,\hbar\,B_k\,y_k\,\mathcal{A}_j\,\eta_j^2\,\xi_i+2\,\gamma\,\hbar\,x_k\,\mathcal{A}_i\,\eta_j\,\xi_i^2-$$
$$6\,\gamma\,\hbar\,B_k\,x_k\,\mathcal{A}_i\,\eta_j\,\xi_i^2+\gamma\,\mathcal{A}_i\,\mathcal{A}_j\,\eta_j^2\,\xi_i^2-4\,\gamma\,B_k\,\mathcal{A}_i\,\mathcal{A}_j\,\eta_j^2\,\xi_i^2+3\,\gamma\,B_k^2\,\mathcal{A}_i\,\mathcal{A}_j\,\eta_j^2\,\xi_i^2\big)\,\epsilon+O[\epsilon]^2\Big]$$

$$\text{R} \to \mathbb{E}_{\{\}\to\{i,j\}}\left[\hbar\,a_j\,b_i,\ \hbar\,x_j\,y_i,\ 1-\frac{1}{4}\left(\gamma\,\hbar^3\,x_j^2\,y_i^2\right)\epsilon+O[\epsilon]^2\right]$$

$$\overline{\text{R}} \to \mathbb{E}_{\{\}\to\{i,j\}}\left[-\hbar\,a_j\,b_i,\ -\frac{\hbar\,x_j\,y_i}{B_i},\ 1-\frac{\left(4\,\hbar^2\,a_j\,B_i\,x_j\,y_i+3\,\gamma\,\hbar^3\,x_j^2\,y_i^2\right)\epsilon}{4\,B_i^2}+O[\epsilon]^2\right]$$

$$\text{P} \to \mathbb{E}_{\{i,j\}\to\{\}}\left[\frac{\alpha_j\,\beta_i}{\hbar},\ \frac{\eta_i\,\xi_j}{\hbar},\ 1+\frac{\gamma\,\eta_i^2\,\xi_j^2\,\epsilon}{4\,\hbar}+O[\epsilon]^2\right]$$

$$\text{aS} \to \mathbb{E}_{\{i\}\to\{i\}}\left[-a_i\,\alpha_i,\ -x_i\,\mathcal{A}_i\,\xi_i,\ 1+\frac{1}{2}\left(-2\,\hbar\,a_i\,x_i\,\mathcal{A}_i\,\xi_i-\gamma\,\hbar\,x_i^2\,\mathcal{A}_i^2\,\xi_i^2\right)\epsilon+O[\epsilon]^2\right]$$

$$\overline{\text{aS}} \to \mathbb{E}_{\{i\}\to\{i\}}\left[-a_i\,\alpha_i,\ -x_i\,\mathcal{A}_i\,\xi_i,\ 1+\frac{1}{2}\left(2\,\gamma\,\hbar\,x_i\,\mathcal{A}_i\,\xi_i-2\,\hbar\,a_i\,x_i\,\mathcal{A}_i\,\xi_i-\gamma\,\hbar\,x_i^2\,\mathcal{A}_i^2\,\xi_i^2\right)\epsilon+O[\epsilon]^2\right]$$

$$\text{bS} \to \mathbb{E}_{\{i\}\to\{i\}}\left[-b_i\,\beta_i,\ -\frac{y_i\,\eta_i}{B_i},\ 1+\frac{\left(-2\,B_i\,y_i\,\beta_i\,\eta_i-\gamma\,\hbar\,y_i^2\,\eta_i^2\right)\epsilon}{2\,B_i^2}+O[\epsilon]^2\right]$$

$$\overline{\text{bS}} \to \mathbb{E}_{\{i\}\to\{i\}}\left[-b_i\,\beta_i,\ -\frac{y_i\,\eta_i}{B_i},\ 1+\frac{\left(2\,\gamma\,\hbar\,B_i\,y_i\,\eta_i-2\,B_i\,y_i\,\beta_i\,\eta_i-\gamma\,\hbar\,y_i^2\,\eta_i^2\right)\epsilon}{2\,B_i^2}+O[\epsilon]^2\right]$$

$$\text{dS} \to \mathbb{E}_{\{i\}\to\{i\}}\Big[-a_i\,\alpha_i-b_i\,\beta_i,\ \frac{-\hbar\,y_i\,\mathcal{A}_i\,\eta_i-\hbar\,B_i\,x_i\,\mathcal{A}_i\,\xi_i+\mathcal{A}_i\,\eta_i\,\xi_i-B_i\,\mathcal{A}_i\,\eta_i\,\xi_i}{\hbar\,B_i},$$
$$1+\frac{1}{4\,\hbar\,B_i^2}\big(4\,\gamma\,\hbar^2\,B_i\,y_i\,\mathcal{A}_i\,\eta_i-4\,\hbar\,B_i\,y_i\,\mathcal{A}_i\,\beta_i\,\eta_i-2\,\gamma\,\hbar^2\,y_i^2\,\mathcal{A}_i^2\,\eta_i^2-4\,\hbar^2\,a_i\,B_i^2\,x_i\,\mathcal{A}_i\,\xi_i-4\,\hbar\,B_i^2\,x_i\,\mathcal{A}_i\,\beta_i\,\xi_i-$$
$$4\,\gamma\,\hbar\,B_i\,\mathcal{A}_i\,\eta_i\,\xi_i+4\,\hbar\,a_i\,B_i\,\mathcal{A}_i\,\eta_i\,\xi_i+4\,\gamma\,\hbar\,B_i^2\,\mathcal{A}_i\,\eta_i\,\xi_i-4\,\gamma\,\hbar^2\,B_i\,x_i\,y_i\,\mathcal{A}_i^2\,\eta_i\,\xi_i+$$
$$4\,B_i\,\mathcal{A}_i\,\beta_i\,\eta_i\,\xi_i-4\,B_i^2\,\mathcal{A}_i\,\beta_i\,\eta_i\,\xi_i+6\,\gamma\,\hbar\,y_i\,\mathcal{A}_i^2\,\eta_i^2\,\xi_i-2\,\gamma\,\hbar\,B_i\,y_i\,\mathcal{A}_i^2\,\eta_i^2\,\xi_i-2\,\gamma\,\hbar^2\,B_i^2\,x_i^2\,\mathcal{A}_i^2\,\xi_i^2+$$
$$6\,\gamma\,\hbar\,B_i\,x_i\,\mathcal{A}_i^2\,\eta_i\,\xi_i^2-2\,\gamma\,\hbar\,B_i^2\,x_i\,\mathcal{A}_i^2\,\eta_i\,\xi_i^2-3\,\gamma\,\mathcal{A}_i^2\,\eta_i^2\,\xi_i^2+4\,\gamma\,B_i\,\mathcal{A}_i^2\,\eta_i^2\,\xi_i^2-\gamma\,B_i^2\,\mathcal{A}_i^2\,\eta_i^2\,\xi_i^2\big)\,\epsilon+O[\epsilon]^2\Big]$$

$$\text{a}\triangle \to \mathbb{E}_{\{i\}\to\{j,k\}}\left[a_j\,\alpha_i+a_k\,\alpha_i,\ x_j\,\xi_i+x_k\,\xi_i,\ 1+\frac{1}{2}\left(-2\,\hbar\,a_j\,x_k\,\xi_i+\gamma\,\hbar\,x_j\,x_k\,\xi_i^2\right)\epsilon+O[\epsilon]^2\right]$$

$$\text{b}\triangle \to \mathbb{E}_{\{i\}\to\{j,k\}}\left[b_j\,\beta_i+b_k\,\beta_i,\ B_k\,y_j\,\eta_i+y_k\,\eta_i,\ 1+\frac{1}{2}\,\gamma\,\hbar\,B_k\,y_j\,y_k\,\eta_i^2\,\epsilon+O[\epsilon]^2\right]$$

$$\text{d}\triangle \to \mathbb{E}_{\{i\}\to\{j,k\}}\Big[a_j\,\alpha_i+a_k\,\alpha_i+b_j\,\beta_i+b_k\,\beta_i,$$
$$y_j\,\eta_i+B_j\,y_k\,\eta_i+x_j\,\xi_i+x_k\,\xi_i,\ 1+\frac{1}{2}\left(\gamma\,\hbar\,B_j\,y_j\,y_k\,\eta_i^2-2\,\hbar\,a_j\,x_k\,\xi_i+\gamma\,\hbar\,x_j\,x_k\,\xi_i^2\right)\epsilon+O[\epsilon]^2\Big]$$

$$\text{C} \to \mathbb{E}_{\{\}\to\{i\}}\left[0,\ 0,\ \sqrt{B_i}-\frac{1}{2}\left(\hbar\,a_i\,\sqrt{B_i}\right)\epsilon+O[\epsilon]^2\right]$$

$$\overline{\text{C}} \to \mathbb{E}_{\{\}\to\{i\}}\left[0,\ 0,\ \frac{1}{\sqrt{B_i}}+\frac{\hbar\,a_i\,\epsilon}{2\,\sqrt{B_i}}+O[\epsilon]^2\right]$$

$$\text{Kink} \to \mathbb{E}_{\{\}\to\{i\}}\left[\hbar\,a_i\,b_i,\ \hbar\,x_i\,y_i,\ \frac{1}{\sqrt{B_i}}+\frac{\left(2\,\hbar\,a_i-\gamma\,\hbar^3\,x_i^2\,y_i^2\right)\epsilon}{4\,\sqrt{B_i}}+O[\epsilon]^2\right]$$

$$\overline{\text{Kink}} \to \mathbb{E}_{\{\}\to\{i\}}\left[-\hbar\,a_i\,b_i,\ -\frac{\hbar\,x_i\,y_i}{B_i},\ \sqrt{B_i}+\frac{\left(-2\,\hbar\,a_i\,B_i^2-4\,\hbar^2\,a_i\,B_i\,x_i\,y_i-3\,\gamma\,\hbar^3\,x_i^2\,y_i^2\right)\epsilon}{4\,B_i^{3/2}}+O[\epsilon]^2\right]$$

$$\text{b2t} \to \mathbb{E}_{\{i\}\to\{i\}}\left[a_i\,\alpha_i-\frac{t_i\,\beta_i}{\gamma},\ y_i\,\eta_i+x_i\,\xi_i,\ 1+\frac{a_i\,\beta_i\,\epsilon}{\gamma}+O[\epsilon]^2\right]$$

$$\text{t2b} \to \mathbb{E}_{\{i\}\to\{i\}}\left[a_i\,\alpha_i-\gamma\,b_i\,\tau_i,\ y_i\,\eta_i+x_i\,\xi_i,\ 1+a_i\,\tau_i\,\epsilon+O[\epsilon]^2\right]$$

Check that on the generators this agrees with our conventions in the handout:

In[∘]:=
```
Timing@
 {{"[a,x]" → ((𝔼{}→{1,2}[0, 0, a2 x1] // am1,2→1)[3] - (𝔼{}→{1,2}[0, 0, a1 x2] // am1,2→1)[3]),
    "[b,y]" → ((𝔼{}→{1,2}[0, 0, y2 b1] // bm1,2→1)[3] - (𝔼{}→{1,2}[0, 0, y1 b2] // bm1,2→1)[3])} /.
   z_1 → z,
  {"Δ[y]" → Last[𝔼{}→{1}[0, 0, y1] ~B1~ bΔ1→1,2],
   "Δ[b]" → Last[𝔼{}→{1}[0, 0, b1] ~B1~ bΔ1→1,2],
   "Δ[a]" → Last[𝔼{}→{1}[0, 0, a1] ~B1~ aΔ1→1,2],
   "Δ[x]" → Last[𝔼{}→{1}[0, 0, x1] ~B1~ aΔ1→1,2]},
  {
    "S(a)" → ((𝔼{}→{1}[0, 0, a1] ~B1~ aS1)[3]),
    "S(x)" → ((𝔼{}→{1}[0, 0, x1] ~B1~ aS1)[3]),
    "S(b)" → ((𝔼{}→{1}[0, 0, b1] ~B1~ bS1)[3]),
    "S(y)" → ((𝔼{}→{1}[0, 0, y1] ~B1~ bS1)[3])
   } /. z_1 → z}
```

Out[∘]= $\{1.57813,$
$\{\{[a,x] \to -x\gamma,\ [b,y] \to -y\epsilon + O[\epsilon]^3\},\ \{\Delta[y] \to (B_2 y_1 + y_2) + O[\epsilon]^3,\ \Delta[b] \to (b_1 + b_2) + O[\epsilon]^3,$
$\Delta[a] \to (a_1 + a_2) + O[\epsilon]^3,\ \Delta[x] \to (x_1 + x_2) - \hbar a_1 x_2 \epsilon + \frac{1}{2}\hbar^2 a_1^2 x_2 \epsilon^2 + O[\epsilon]^3\},\ \{S(a) \to -a + O[\epsilon]^3,$
$S(x) \to -x - a x \hbar \epsilon - \frac{1}{2}(a^2 x \hbar^2)\epsilon^2 + O[\epsilon]^3,\ S(b) \to -b + O[\epsilon]^3,\ S(y) \to -\frac{y}{B} + O[\epsilon]^3\}\}\}$

Hopf algebra axioms on both sides separately.

Associativity of am and bm:

In[∘]:=
```
Timing@Block[{$k = 3},
  HL /@ {(am1,2→1 // am1,3→1) ≡ (am2,3→2 // am1,2→1), (bm1,2→1 // bm1,3→1) ≡ (bm2,3→2 // bm1,2→1)}
 ]
```

Out[∘]= {0.125, {True, True}}

R and *P* are inverses:

In[∘]:=
```
Timing@Block[{$k = 3}, {Ri,j, Pi,k, HL[(Ri,j // Pi,k) ≡ 𝔼{k}→{j}[aj αk, xj ξk, 1]]}]
```

Out[∘]= $\{0.09375, \{\mathbb{E}_{\{\}\to\{i,j\}}[\hbar a_j b_i, \hbar x_j y_i, 1 - \frac{1}{4}(\gamma\hbar^3 x_j^2 y_i^2)\epsilon + (\frac{1}{9}\gamma^2\hbar^5 x_j^3 y_i^3 + \frac{1}{32}\gamma^2\hbar^6 x_j^4 y_i^4)\epsilon^2 +$
$\frac{1}{1152}(24\gamma^3\hbar^5 x_j^2 y_i^2 - 72\gamma^3\hbar^7 x_j^4 y_i^4 - 32\gamma^3\hbar^8 x_j^5 y_i^5 - 3\gamma^3\hbar^9 x_j^6 y_i^6)\epsilon^3 + O[\epsilon]^4],$
$\mathbb{E}_{\{i,k\}\to\{\}}[\frac{\alpha_k\beta_i}{\hbar}, \frac{\eta_i\xi_k}{\hbar}, 1 + \frac{\gamma\eta_i^2\xi_k^2\epsilon}{4\hbar} + \frac{(36\gamma^2\hbar^2\eta_i^2\xi_k^2 + 40\gamma^2\hbar\eta_i^3\xi_k^3 + 9\gamma^2\eta_i^4\xi_k^4)\epsilon^2}{288\hbar^2} - \frac{1}{1152\hbar^3}$
$(-48\gamma^3\hbar^4\eta_i^2\xi_k^2 - 192\gamma^3\hbar^3\eta_i^3\xi_k^3 - 156\gamma^3\hbar^2\eta_i^4\xi_k^4 - 40\gamma^3\hbar\eta_i^5\xi_k^5 - 3\gamma^3\eta_i^6\xi_k^6)\epsilon^3 + O[\epsilon]^4],\ \text{True}\}\}$

as and $\overline{\text{aS}}$ are inverses, bs and $\overline{\text{bS}}$ are inverses:

In[∘]:=
```
Timing[HL /@ {(aS̄1 // aS1) ≡ 𝔼{1}→{1}[a1 α1, x1 ξ1, 1], (bS̄1 // bS1) ≡ 𝔼{1}→{1}[b1 β1, y1 η1, 1]}]
```

Out[∘]= {0.359375, {True, True}}

(co)-associativity on both sides

```
In[∘]:= Timing[
   HL /@ {(aΔ1→1,2 // aΔ2→2,3) ≡ (aΔ1→1,3 // aΔ1→1,2), (bΔ1→1,2 // bΔ2→2,3) ≡ (bΔ1→1,3 // bΔ1→1,2),
     (am1,2→1 // am1,3→1) ≡ (am2,3→2 // am1,2→1), (bm1,2→1 // bm1,3→1) ≡ (bm2,3→2 // bm1,2→1)}]
Out[∘]= {0.390625, {True, True, True, True}}
```

Δ is an algebra morphism

```
In[∘]:= Timing[HL /@ {(am1,2→1 // aΔ1→1,2) ≡ ((aΔ1→1,3 aΔ2→2,4) // (am3,4→2 am1,2→1)),
     (bm1,2→1 // bΔ1→1,2) ≡ ((bΔ1→1,3 bΔ2→2,4) // (bm3,4→2 bm1,2→1))}]
Out[∘]= {0.65625, {True, True}}
```

An explicit formula for aS_i

```
In[∘]:= Timing@Block[{$k = 4}, HL[aSi ≡ (𝔼{i}→{i,j}[-αi aj, -ξi xi,
        Sum[Expand[(ⅇ^(ξi xi) (-ℏ γ ϵ)^k)/(2^k k!) Nest[Expand[xi^2 ∂{xi,2} #] &, ⅇ^(-ξi ⅇ^(ℏ ϵ ai) xi), k]], {k, 0, $k}]]$k //
      ami,j→i)]]
Out[∘]= {2.96875, True}
```

S is convolution inverse of id

```
In[∘]:= Timing[HL[# ≡ 𝔼{1}→{1}[0, 0, 1]] & /@ {
    (aΔ1→1,2 ~ B1 ~ aS1) ~ B1,2 ~ am1,2→1, (aΔ1→1,2 ~ B2 ~ aS2) ~ B1,2 ~ am1,2→1,
    (bΔ1→1,2 ~ B1 ~ bS1) ~ B1,2 ~ bm1,2→1, (bΔ1→1,2 ~ B2 ~ bS2) ~ B1,2 ~ bm1,2→1}]
Out[∘]= {0.5625, {True, True, True, True}}
```

But not with the opposite product:

```
In[∘]:= Timing[Short[# ≡ 𝔼{1}→{1}[0, 0, 1]] & /@ {
    (aΔ1→1,2 ~ B1 ~ aS1) ~ B1,2 ~ am2,1→1, (aΔ1→1,2 ~ B2 ~ aS2) ~ B1,2 ~ am2,1→1,
    (bΔ1→1,2 ~ B1 ~ bS1) ~ B1,2 ~ bm2,1→1, (bΔ1→1,2 ~ B2 ~ bS2) ~ B1,2 ~ bm2,1→1}]
```

Out[∘]= $\left\{0.625, \left\{\frac{1}{2}\left(-2\,\gamma\,\epsilon\,\hbar\,x_1\,\mathcal{A}_1\,\xi_1 + \gamma^2\,\epsilon^2\,\hbar^2\,x_1\,\mathcal{A}_1\,\xi_1 - 2\,\gamma \ll 4 \gg \mathcal{A}_1\,\xi_1 + 2\,\gamma^2\,\epsilon^2\,\hbar^2\,x_1^2\,\mathcal{A}_1^2\,\xi_1^2\right) == 0,\right.\right.$

$\frac{1}{2}\left(-2\,\gamma\,\epsilon\,\hbar\,x_1\,\xi_1 - \gamma^2\,\epsilon^2\,\hbar^2\,x_1\,\xi_1 + 2\,\gamma^2\,\epsilon^2\,\hbar^2\,x_1^2\,\xi_1^2\right) == 0,$

$\left.\left.\frac{1}{2}\left(-2\,\gamma\,\epsilon\,\hbar\,y_1\,\eta_1 - \gamma^2\,\epsilon^2\,\hbar^2\,y_1\,\eta_1 + 2\,\gamma^2\,\epsilon^2\,\hbar^2\,y_1^2\,\eta_1^2\right) == 0,\ \frac{-2\,\gamma\,\epsilon\,\hbar\,B_1\,y_1\,\eta_1 + \ll 4 \gg}{2\,B_1^2} == 0\right\}\right\}$

S is an algebra anti-(co)morphism

```
In[∘]:= Timing[HL /@ {am1,2→1 ~ B1 ~ aS1 ≡ (aS1 aS2) ~ B1,2 ~ am2,1→1, bm1,2→1 ~ B1 ~ bS1 ≡ (bS1 bS2) ~ B1,2 ~ bm2,1→1,
    aS1 ~ B1 ~ aΔ1→1,2 ≡ aΔ1→2,1 ~ B1,2 ~ (aS1 aS2), bS1 ~ B1 ~ bΔ1→1,2 ≡ bΔ1→2,1 ~ B1,2 ~ (bS1 bS2)}]
Out[∘]= {0.859375, {True, True, True, True}}
```

Pairing axioms

In[]:=
```
Timing[HL /@ {(bm1,2→1 𝔼{3}→{3}[α3 a3, ξ3 x3, 1]) ~B1,3~P1,3 ≡
    (𝔼{1}→{1}[β1 b1, η1 y1, 1] 𝔼{2}→{2}[β2 b2, η2 y2, 1] aΔ3→4,5) ~B1,4~P1,4~B2,5~P2,5,
   (bΔ1→1,2 𝔼{3}→{3}[α3 a3, ξ3 x3, 1] 𝔼{4}→{4}[α4 a4, ξ4 x4, 1]) ~B1,3~P1,3~B2,4~P2,4 ≡
    (𝔼{1}→{1}[β1 b1, η1 y1, 1] am3,4→3) ~B1,3~P1,3 }]
```

Out[]= {0.375, {True, True}}

In[]:=
```
Timing[HL /@ {((bS1 𝔼{2}→{2}[α2 a2, ξ2 x2, 1]) // P1,2) ≡ ((𝔼{1}→{1}[β1 b1, η1 y1, 1] aS2) // P1,2),
   (b̄S̄1 𝔼{2}→{2}[α2 a2, ξ2 x2, 1]) ~B1,2~P1,2 ≡ (𝔼{1}→{1}[β1 b1, η1 y1, 1] āS̄2) ~B1,2~P1,2}]
```

Out[]= {0.28125, {True, True}}

Tests for the double.

Check the double formulas on the generators agree with SL2Portfolio.pdf:

In[]:=
```
Timing@{{
     "[a,y]" →
      ((𝔼{}→{1,2}[0, 0, y2 a1] ~B1,2~dm1,2→1) [3] - (𝔼{}→{1,2}[0, 0, y1 a2] ~B1,2~dm1,2→1) [3]),
     "[b,x]"  → ((𝔼{}→{1,2}[0, 0, x2 b1] ~B1,2~dm1,2→1) [3] -
        (𝔼{}→{1,2}[0, 0, x1 b2] ~B1,2~dm1,2→1) [3]),
     "xy-qyx" → ((𝔼{}→{1,2}[0, 0, x1 y2] ~B1,2~dm1,2→1) [3] -
        (1 + ϵ) (𝔼{}→{1,2}[0, 0, y1 x2] ~B1,2~dm1,2→1) [3])
    } /. {z_1 → z} // Expand // Factor,
  {
   "Δ(a)" → ((𝔼{}→{1}[0, 0, a1] ~B1~dΔ1→1,2) [3]),
   "Δ(x)" → ((𝔼{}→{1}[0, 0, x1] ~B1~dΔ1→1,2) [3]),
   "Δ(b)" → ((𝔼{}→{1}[0, 0, b1] ~B1~dΔ1→1,2) [3]),
   "Δ(y)" → ((𝔼{}→{1}[0, 0, y1] ~B1~dΔ1→1,2) [3])
   } // Simplify,
  {
    "S(a)" → ((𝔼{}→{1}[0, 0, a1] ~B1~dS1) [3]),
    "S(x)" → ((𝔼{}→{1}[0, 0, x1] ~B1~dS1) [3]),
    "S(b)" → ((𝔼{}→{1}[0, 0, b1] ~B1~dS1) [3]),
    "S(y)" → ((𝔼{}→{1}[0, 0, y1] ~B1~dS1) [3])
   } /. {z_1 → z} // Simplify
 }
```

Out[]=
$$\Big\{7.1875, \Big\{\Big\{[a,y] \to -y\,\gamma + O[\epsilon]^3,\ [b,x] \to x\,\epsilon + O[\epsilon]^3,$$
$$xy\text{-}qyx \to \left(-x\,y + \frac{1 - B + x\,y\,\hbar}{\hbar}\right) + (a\,B - x\,y + x\,y\,\gamma\,\hbar)\,\epsilon + \frac{1}{2}\left(-a^2\,B\,\hbar + x\,y\,\gamma^2\,\hbar^2\right)\epsilon^2 + O[\epsilon]^3\Big\},$$
$$\Big\{\Delta(a) \to (a_1 + a_2) + O[\epsilon]^3,\ \Delta(x) \to (x_1 + x_2) - \hbar\,a_1\,x_2\,\epsilon + \frac{1}{2}\hbar^2\,a_1^2\,x_2\,\epsilon^2 + O[\epsilon]^3,$$
$$\Delta(b) \to (b_1 + b_2) + O[\epsilon]^3,\ \Delta(y) \to (y_1 + B_1\,y_2) + O[\epsilon]^3\Big\},$$
$$\Big\{S(a) \to -a + O[\epsilon]^3,\ S(x) \to -x - a\,x\,\hbar\,\epsilon - \frac{1}{2}\left(a^2\,x\,\hbar^2\right)\epsilon^2 + O[\epsilon]^3,$$
$$S(b) \to -b + O[\epsilon]^3,\ S(y) \to -\frac{y}{B} + \frac{y\,\gamma\,\hbar\,\epsilon}{B} - \frac{\left(y\,\gamma^2\,\hbar^2\right)\epsilon^2}{2\,B} + O[\epsilon]^3\Big\}\Big\}\Big\}$$

(co)-associativity

```
In[∘]:= Timing[
         HL /@ {(dΔ1→1,2 // dΔ2→2,3) ≡ (dΔ1→1,3 // dΔ1→1,2), (dm1,2→1 // dm1,3→1) ≡ (dm2,3→2 // dm1,2->1)}]
Out[∘]= {6.04688, {True, True}}
```

Δ is an algebra morphism

```
In[∘]:= Timing@HL[dm1,2→1 ~ B1 ~ dΔ1→1,2 ≡ (dΔ1→1,3 dΔ2→2,4) ~ B1,2,3,4 ~ (dm3,4→2 dm1,2→1)]
Out[∘]= {6.125, True}
```

S_2 inverts R, but not S_1:

```
In[∘]:= Timing@{R1,2 ~ B1 ~ dS1 ≡ R̄1,2, HL[R1,2 ~ B2 ~ dS2 ≡ R̄1,2]}
```

Out[∘]= $\left\{0.65625, \left\{\frac{1}{4 B_1^3}\left(4 \gamma \epsilon \hbar^2 B_1^2 x_2 y_1 - 2 \gamma^2 \epsilon^2 \hbar^3 B_1^2 x_2 y_1 + 4 \gamma \epsilon^2 \hbar^3 a_2 B_1^2 x_2 y_1 + 8 \gamma^2 \epsilon^2 \hbar^4 B_1 x_2^2 y_1^2 - 4 \gamma \epsilon^2 \hbar^4 a_2 B_1 x_2^2 y_1^2 - 3 \gamma^2 \epsilon^2 \hbar^5 x_2^3 y_1^3\right) == 0, \text{True}\right\}\right\}$

S is convolution inverse of id

```
In[∘]:= Timing[HL[# ≡ 𝔼{1}→{1}[0, 0, 1]] & /@
          {(dΔ1→1,2 ~ B1 ~ dS1) ~ B1,2 ~ dm1,2→1, (dΔ1→1,2 ~ B2 ~ dS2) // dm1,2→1}]
Out[∘]= {8.29688, {True, True}}
```

S is a (co)-algebra anti-morphism

```
In[∘]:= Timing[HL /@
         Expand /@ {dm1,2→1 ~ B1 ~ dS1 ≡ (dS1 dS2) ~ B1,2 ~ dm2,1→1, dS1 ~ B1 ~ dΔ1→1,2 ≡ dΔ1→2,1 ~ B1,2 ~ (dS1 dS2)}]
Out[∘]= {15.5938, {True, True}}
```

Quasi-triangular axiom 1:

```
In[∘]:= Timing@HL[R1,2 ~ B1 ~ dΔ1->1,3 ≡ (R1,4 R3,2) ~ B2,4 ~ dm2,4→2]
Out[∘]= {0.546875, True}
```

Quasi-triangular axiom 2:

```
In[∘]:= Timing@HL[((dΔ1->1,2 R3,4) ~ B1,2,3,4 ~ (dm1,3→1 dm2,4→2)) ≡ ((dΔ1→2,1 R3,4) ~ B1,2,3,4 ~ (dm3,1→1 dm4,2→2))]
Out[∘]= {5.75, True}
```

The Drinfel'd element inverse property, $(u_1\, \overline{u}_2) \sim B_{1,2} \sim \mathrm{dm}_{1,2\to 1} \equiv \mathbb{E}[0, 0, 1]$:

```
In[∘]:= Timing@HL[((R1,2 ~ B1 ~ dS1 ~ B1,2 ~ dm2,1→i) (R1,2 ~ B2 ~ dS2 ~ B2 ~ dS2 ~ B1,2 ~ dm2,1→j)) ~ Bi,j ~ dmi,j→i ≡
          𝔼{}→{i}[0, 0, 1]]
Out[∘]= {2.73438, True}
```

The ribbon element v satisfies $v^2 = S(u)\, u$. The spinner C=uv^{-1}. It is convenient to compute $z = S(u)\, u^{-1}$ which is something easy.

```
In[]:= Timing@Block[{$k = 2},
  (((R1,2 ~ B1 ~ dS1 ~ B1,2 ~ dm2,1→i) ~ Bi ~ dSi) (R1,2 ~ B2 ~ dS2 ~ B2 ~ dS2 ~ B1,2 ~ dm2,1→j)) ~ Bi,j ~ dmi,j→i]
```

Out[]= $\{3.21875, \mathbb{E}_{\{\}\to\{i\}}[0, 0, \frac{1}{B_i} + \frac{\hbar a_i \epsilon}{B_i} + \frac{\hbar^2 a_i^2 \epsilon^2}{2 B_i} + O[\epsilon]^3]\}$

```
In[]:= Timing@Block[{$k = 2}, HL /@ {(Ci C̄j) ~ Bi,j ~ dmi,j→i ≡ 𝔼{}→{i}[0, 0, 1], (C̄i C̄j) ~ Bi,j ~ dmi,j→i ≡
   (((R1,2 ~ B1 ~ dS1 ~ B1,2 ~ dm2,1→i) ~ Bi ~ dSi) (R1,2 ~ B2 ~ dS2 ~ B2 ~ dS2 ~ B1,2 ~ dm2,1→j)) ~ Bi,j ~ dmi,j→i}]
```

Out[]= {3.76563, {True, True}}

Reidemeister 2:

```
In[]:= Timing[HL[# ≡ 𝔼{}→{1,2}[0, 0, 1]] & /@
  {(R̄1,2 R3,4) ~ B1,2,3,4 ~ (dm1,3→1 dm2,4→2), (R1,2 R̄3,4) ~ B1,2,3,4 ~ (dm1,3→1 dm2,4→2)}]
```

Out[]= {4.73438, {True, True}}

Cyclic Reidemeister 2:

```
In[]:= Timing@HL[(R1,4 R̄5,2 C̄3) ~ B2,4 ~ dm2,4→2 ~ B1,3 ~ dm1,3→1 ~ B1,5 ~ dm1,5→1 ≡ C̄1 𝔼{}→{2}[0, 0, 1]]
```

Out[]= {1.96875, True}

Reidemeister 3:

```
In[]:= Timing@HL[((R1,2 R4,3 R5,6) ~ B1,4 ~ dm1,4→1 ~ B2,5 ~ dm2,5→2 ~ B3,6 ~ dm3,6→3) ≡
  ((R1,6 R2,3 R4,5) ~ B1,4 ~ dm1,4→1 ~ B2,5 ~ dm2,5→2 ~ B3,6 ~ dm3,6→3)]
```

Out[]= {4.26563, True}

Relations between the four kinks:

```
In[]:= Timing[HL /@ {Kinki ≡ (R3,1 C2) ~ B1,2 ~ dm1,2→1 ~ B1,3 ~ dm1,3→i,
  K̄inkj ≡ (R̄3,1 C̄2) ~ B1,2 ~ dm1,2→1 ~ B1,3 ~ dm1,3→j, (Kinki K̄inkj) ~ Bi,j ~ dmi,j→1 ≡ 𝔼{}→{1}[0, 0, 1]}]
```

Out[]= {4.09375, {True, True, True}}

Trefoil

The Trefoil

Trefoil

In[]:=
```
$k = 2; Z = kR1,5 kR6,2 kR3,7 kC4 kKink8 kKink9 kKink10;
Do[Z = Z~B1,r~km1,r→1, {r, 2, 10}];
Simplify /@ Z /. v_1 :→ v
```

$\overline{kC}_4\ \overline{kKink}_8\ \overline{kKink}_9\ \overline{kKink}_{10}$

Trefoil

Out[]=
$$\mathbb{E}_{\{\}\to\{1\}}\Big[0,\ 0,\ \frac{T}{1-T+T^2}+\left(T\hbar\left(2a\left(-1+T-T^3+T^4\right)+T\left(-1+2T-3T^2+2T^3\right)\gamma-2\left(1+T^3\right)xy\gamma\hbar\right)\epsilon\right)\Big/\left(1-T+T^2\right)^3+\frac{1}{2\left(1-T+T^2\right)^5}T\hbar^2\Big(4a^2\left(1-T+T^2\right)^2\left(1+T-6T^2+T^3+T^4\right)+4a\left(1-T+T^2\right)\gamma\left(T\left(2-5T+8T^2-7T^3-2T^4+2T^5\right)-2\left(-1-2T+5T^2-4T^3+T^4+2T^5\right)xy\hbar\right)+\gamma^2\Big(T\left(1-2T+4T^2-2T^3+6T^5-11T^6+4T^7\right)+4\left(-1+2T+T^3+T^4+2T^6-T^7\right)xy\hbar+6\left(1-T+T^2\right)^2\left(1+3T+T^2\right)x^2y^2\hbar^2\Big)\Big)\epsilon^2+O[\epsilon]^3\Big]$$